

The Functional Correspondence Applied: An Implementation of a Semantics Transformer

Maciej Buszka

Instytut Informatyki UW

18.11.2020

The Functional Correspondence

and an introduction to the meta-language

The Functional Correspondence

- A method for deriving abstract machines
- Takes a definitional interpreter
- Produces an encoding of an abstract machine
- Introduced by Danvy et al. in *A functional correspondence between evaluators and abstract machines*
- Based on two source-to-source transformations
 - ▶ Translation to continuation-passing style (CPS)
 - ▶ Defunctionalization

Meta-language by Example: CBV λ -calculus

```
(def-data Term
  String          ;; variable
  {Abs String Term} ;; lambda abstraction
  {App Term Term}) ;; application

;; Environment is encoded as a partial function

(def extend (env x v) ... ) ;; extends env at x with value v

(def eval (env term)
  (match term
    ([String x] (env x))
    ({Abs x body} (fun (v) (eval (extend env x v) body)))
    ({App fn arg} ((eval env fn) (eval env arg)))))
```

Transformation Steps We Aim For

(Also a reminder of how functional correspondence works)

After CPS Translation

```
(def eval (env term cont)
  (match term
    ([String x] (cont (env x)))
    ({Abs x body}
     (cont (fun (v cont)
               (eval (extend env x v) body cont)))))
    ({App fn arg}
     (eval env fn (fun (fn)
                       (eval env arg (fun (arg) (fn arg cont))))))))
```

After Defunctionalization

```
(def eval (env term cont)
  (match term
    ([String x]   (continue cont (env x)))
    ({Abs x body} (continue cont {Fun body env x}))
    ({App fn arg} (eval env fn {App1 arg cont env}))))

(def apply (fn v cont)
  (match fn ({Fun body env x}
    (eval (extend env x v) body cont))))

(def continue (cont val)
  (match cont
    ({App1 arg cont env} (eval env arg {App2 cont val}))
    ({App2 cont fn}      (apply fn val cont))
    ({Halt}              val)))
```

The Result

- An abstract machine encoded as mutually tail-recursive functions
- Uses abstract environment (encoded as a function)
 - ▶ extend is still higher-order
 - ▶ extend and functions representing environment remain in direct style

Desired Features

- Metalanguage expressivity
 - ▶ Mutually recursive, anonymous and higher-order functions
 - ▶ Records and pattern matching
 - ▶ Dynamic (strong) typing
- Automation
 - ▶ Function space detection
 - ▶ CPS and defunctionalization
 - ▶ Sensible name generation
- User's control over transformation
 - ▶ Naming of records for defunctionalized lambdas, apply functions
 - ▶ Disabling defunctionalization and/or cps transformation of chosen functions

Why Control is Needed

```
(def extend (env y v cont1) (continue cont1 {Ext env v y}))  
(def lookup (fn1 x cont2) ... (continue2 ...))  
  
(def eval (env term cont3)  
  (match term  
    ([String x] (lookup env x cont3))  
    ({Abs x body} (continue2 cont3 {Fun body env x}))  
    ({App fn arg} (eval env fn {App1 arg cont3 env}))))  
  
(def continue2 (fn4 var3)  
  (match fn4  
    ({App2 cont3 var2} (apply var2 var3 cont3))  
    ({App1 arg cont3 env} (eval env arg {App2 cont3 var3}))  
    ({Halt } var3)))  
  
(def apply (fn2 v cont4) ... (extend ...))
```

Automating the Functional Correspondence

(A particular approach)

In a Nutshell

- Transformation to ANF
- Control flow analysis
- Transformation to CPS
- Control flow analysis
- Defunctionalization
- Inlining administrative let-bindings

Transformation to ANF

- Assumes left-to-right call-by-value semantics
- Let-bind every intermediate result
- Eases program analysis and further transformations
- Preserves tail calls

Abstract syntax of the meta-language

$x, y, z \in Var$ $r \in StructName$

$s \in String$ $b \in Int \cup Boolean \cup String$

$Tp \ni tp ::= String \mid Integer \mid Boolean$

$Pattern \ni p ::= x \mid b \mid _ \mid \{r \ p \dots\} \mid [tp \ x]$

$Term \ni t ::= x \mid b \mid (\text{fun } (x \dots) \ t) \mid (t \ t \dots) \mid \{r \ t \dots\}$
 $\mid (\text{let } p \ t \ t) \mid (\text{match } t \ (p \ t) \dots) \mid (\text{error } s)$

Transformation to ANF

Abstract syntax in ANF

$$\begin{aligned} Com \ni c &::= x \mid b \mid (\text{fun } (x \dots) e) \mid (x \ x \dots) \\ &\quad \mid \{r \ x \dots\} \mid (\text{match } x \ (p \ e) \dots) \\ Anf \ni e &::= c \mid (\text{let } p \ c \ e) \mid (\text{error } s) \end{aligned}$$

ANF translation

$$\begin{aligned} \llbracket \cdot \rrbracket \cdot &: Term \times (Com \rightarrow Anf) \rightarrow Anf \\ \llbracket a \rrbracket k &= k \ a \\ \llbracket (\text{fun } (x \dots) t) \rrbracket k &= k \ (\text{fun } (x \dots) \llbracket t \rrbracket id) \\ \llbracket (t_f \ t_{arg} \dots) \rrbracket k &= \llbracket t_f \rrbracket [\lambda x_f. \llbracket t_{arg} \dots \rrbracket_s \lambda(x_{arg} \dots). k \ (x_f \ x_{arg} \dots)]_a \\ \llbracket (\text{let } x \ t_1 \ t_2) \rrbracket k &= \llbracket t_1 \rrbracket \lambda c_1. (\text{let } x \ c_1 \llbracket t_2 \rrbracket k) \\ \llbracket \{r \ t \dots\} \rrbracket k &= \llbracket t \dots \rrbracket_s \lambda(x \dots). k \ \{r \ x \dots\} \\ \llbracket (\text{match } t \ (p \ t_b)) \rrbracket k &= \llbracket t \rrbracket [\lambda x. k \ (\text{match } x \ (p \ \llbracket t_b \rrbracket id)]_a \\ \llbracket (\text{error } s) \rrbracket _ &= (\text{error } s) \end{aligned}$$

Transformation to ANF

Extending atomic continuation

$$\begin{aligned} [\cdot]_a \cdot &: (Var \rightarrow Anf) \rightarrow Com \rightarrow Anf \\ [k]_a x &= k x \\ [k]_a c &= (\text{let } x \text{ } c \text{ } (k x)) \end{aligned}$$

Sequencing multiple terms

$$\begin{aligned} \llbracket \cdot \rrbracket_s \cdot &: Term^* \times (Var^* \rightarrow Anf) \rightarrow Anf \\ \llbracket t \dots \rrbracket_s k &= \llbracket t \dots \rrbracket_s^\epsilon \\ \llbracket \epsilon \rrbracket_s^{x \dots} k &= k(x \dots) \\ \llbracket t \ t_r \dots \rrbracket_s^{x_{acc} \dots} k &= \llbracket t \rrbracket [\lambda x. \llbracket t_r \dots \rrbracket_s^{x_{acc} \dots x}]_a \end{aligned}$$

After ANF Transformation

```
(def eval (env term)
  (match term
    ([String x] (env x))
    ({Abs x body}
     (fun (v)
       (let var1 (extend env x v))
       (eval var1 body)))
    ({App fn arg}
     (let var2 (eval env fn))
     (let var3 (eval env arg))
     (var2 var3))))
```


Control Flow Analysis

- For each expression in a program, find the over-approximation of the set of functions it may evaluate to
- Detects function spaces for defunctionalization
- Enables selective cps translation
- Textbook approaches:
 - ▶ Constraint systems
 - ▶ Annotated type systems

Abstracting Abstract Machines

- Begin with a CESK* machine extended with pattern matching (Control Environment Store Kontinuation pointer)
- Because the program is in ANF:
 - ▶ There are only two types of continuations
 - ▶ Every proper subexpression will allocate a value in the store
- Provide finite approximations for base types
- Let the store map to sets of value approximations
- Compute the set of configurations reachable from the initial one
- Now for every variable in function position the store contains a set of values to which it may evaluate
- Methodology introduced by Van Horn et al. in *Abstracting Abstract Machines*

AAM – Performance Considerations

- Use a single shared store (corresponds to widening)
- Trim environment in closures and continuations
- Sufficiently fast even with naive fixed-point iteration

Selective Translation to CPS

- Allows the user to specify which functions should be left in direct style
- Functions in direct style may call CPS ones and vice versa
- Quite simple since the program is already in ANF
- Designed not to duplicate code
- Produces a program in ANF

CPS Annotations

```
(def init #:atomic (x) (error "empty environment"))
```

```
(def extend #:atomic (env y v)  
  (fun #:atomic (x) (match (eq? x y)  
    (#t v)  
    (#f (env x))))))
```

Selective Translation to CPS

Translation to CPS

$$\begin{aligned}\llbracket x \rrbracket_c k &= (k \ x) \\ \llbracket b \rrbracket_c k &= (\text{let } x \ b \ (k \ x)) \\ \llbracket \{r \ x \dots\} \rrbracket_c k &= (\text{let } y \ \{r \ x \dots\} \ (k \ y)) \\ \llbracket (\text{fun } \#:\text{atomic} \ (x \dots) e) \rrbracket_c k &= (\text{let } y \ (\text{fun } (x \dots) \llbracket e \rrbracket_d) \ (k \ y)) \\ \llbracket (\text{fun } (x \dots) e) \rrbracket_c k &= (\text{let } y \ (\text{fun } (x \dots k') \llbracket e \rrbracket_c k') \ (k \ y)) \\ \llbracket (f' \ x \dots) \rrbracket_c k &= \begin{cases} (f \ x \dots k) & \text{when } \text{noneAtomic}(l) \\ (\text{let } y \ (f \ x \dots) \ (k \ y)) & \text{when } \text{allAtomic}(l) \end{cases} \\ \llbracket (\text{match } x \ (p \ e) \dots) \rrbracket_c k &= (\text{match } x \ (p \ \llbracket e \rrbracket_c k) \dots) \\ \llbracket (\text{let } x \ c \ e) \rrbracket_c k &= \begin{cases} (\text{let } x \ \llbracket c \rrbracket_d \ \llbracket e \rrbracket_c k) & \text{when } \text{trivial}(c) \\ (\text{let } k' \ (\text{fun } (x) \ \llbracket e \rrbracket_c k) \ \llbracket c \rrbracket_c k') & \text{otherwise} \end{cases} \\ \llbracket (\text{error } s) \rrbracket_c k &= (\text{error } s)\end{aligned}$$

Selective Translation to CPS

Translation for terms left in direct style

$$\begin{aligned}\llbracket x \rrbracket_d &= x \\ \llbracket b \rrbracket_d &= b \\ \llbracket \{r \ x \dots\} \rrbracket_d &= \{r \ x \dots\} \\ \llbracket (\text{fun } \#:\text{atomic} \ (x \dots) e) \rrbracket_d &= (\text{fun } (x \dots) \llbracket e \rrbracket_d) \\ \llbracket (\text{fun } (x \dots) e) \rrbracket_d &= (\text{fun } (x \dots k') \llbracket e \rrbracket_c k') \\ \llbracket (f' \ x \dots) \rrbracket_d &= \begin{cases} (f \ x \dots) & \text{when } allAtomic(I) \\ (\text{let } k \ (\text{fun } (y) \ y) \ (f \ x \dots k)) & \text{when } noneAtomic(I) \end{cases} \\ \llbracket (\text{match } x \ (p \ e) \dots) \rrbracket_d &= (\text{match } x \ (p \ \llbracket e \rrbracket_d) \dots) \\ \llbracket (\text{let } x \ (f' \ y \dots) \ e) \rrbracket_d &= \begin{aligned} &(\text{let } k \ (\text{fun } (z) \ z) \\ &(\text{let } x \ (f \ y \dots k)) \llbracket e \rrbracket_d) \end{aligned} \quad \text{when } noneAtomic(I) \\ \llbracket (\text{let } x \ c \ e) \rrbracket_d &= (\text{let } x \ \llbracket c \rrbracket_d \ \llbracket e \rrbracket_d) \\ \llbracket (\text{error } s) \rrbracket_d &= (\text{error } s) \end{aligned}$$

After CPS Translation

```
(def eval (env term cont)
  (match term
    ([String x]
     (let val (env x))
     (cont val)))
    ({Abs x body}
     (let val1
       (fun (v cont1)
         (let var1 (extend env x v))
         (eval var1 body cont1))))
     (cont val1)))
    ({App fn arg}
     (eval env fn (fun (var2)
                       (eval env arg (fun (var3)
                                           (var2 var3 cont))))))))))
```


Selective Defunctionalization

- Control flow analysis provides all the necessary information
- Trade-offs and gotchas
 - ▶ Direct vs indirect calls to top-level functions
 - ▶ Primitive operations
- Record name generation
 - ▶ For anonymous functions: let the user decide
 - ▶ For continuations: use the constructor of current branch (heuristic)
- Result is no longer in ANF

Defunctionalization Annotations

```
(def init #:no-defun (x) (error "empty environment"))
```

```
(def extend (env y v)
  (fun #:no-defun (x) (match (eq? x y)
    (#t v)
    (#f (env x))))))
```

```
(def eval (env term)
  (match term
    ([String x] (env x))
    ({Abs x body} (fun #:name Fun #:apply apply (v) (eval (extend env x v) body)))
    ({App fn arg} ((eval env fn) (eval env arg)))))
```

Selective Defunctionalization

$$\begin{aligned}
 \llbracket x \rrbracket &= \begin{cases} \{\text{Prim}_x\} & \text{when } \text{primOp}(x) \\ \{\text{Top}_x\} & \text{when } \text{topLevel}(x) \wedge \text{defun}(x) \\ x & \text{otherwise} \end{cases} \\
 \llbracket b \rrbracket &= b \\
 \llbracket \{r \ x \dots\} \rrbracket &= \{r \ \llbracket x \rrbracket \dots\} \\
 \llbracket (\text{fun } (x \dots) e)^I \rrbracket &= \begin{cases} \{\text{Fun}_I \ \text{fvs}(e)\} & \text{when } \text{defun}(I) \\ (\text{fun } (x \dots) \llbracket e \rrbracket) & \text{otherwise} \end{cases} \\
 \llbracket (f' \ x \dots)^I \rrbracket &= \begin{cases} (f \ \llbracket x \rrbracket \dots) & \text{when } \text{primOp}(f) \vee \\ & \text{topLevel}(f) \\ (\text{apply}_I \ f \ \llbracket x \rrbracket \dots) & \text{else when } \text{allDefun}(I') \\ (f \ \llbracket x \rrbracket \dots) & \text{when } \text{noneDefun}(I') \end{cases} \\
 \llbracket (\text{match } x \ (p \ e) \dots) \rrbracket &= (\text{match } x \ (p \ \llbracket e \rrbracket) \dots) \\
 \llbracket (\text{let } x \ c \ e) \rrbracket &= (\text{let } x \ \llbracket c \rrbracket \ \llbracket e \rrbracket) \\
 \llbracket (\text{error } s) \rrbracket &= (\text{error } s)
 \end{aligned}$$

Selective Defunctionalization

Apply function generation

$$\begin{aligned} \text{mkBranch}(x \dots, \delta) &= (\{\text{Prim}_\delta\} (\delta \ x \dots)) \\ \text{mkBranch}(x \dots, (\text{def } f \ (y \dots) \ e)) &= (\{\text{Top}_f\} (f \ x \dots)) \\ \text{mkBranch}(x \dots, (\text{fun } (y \dots) \ e)^I) &= (\{\text{Fun}_I \ fvs(e)\} \llbracket e \rrbracket [y \mapsto x \dots]) \\ \text{mkApply}(I, fn \dots) &= (\text{def apply}_I \ (f \ x \dots) \\ &\quad (\text{match } f \\ &\quad \quad \text{mkBranch}(x \dots, fn) \dots)) \end{aligned}$$

After Defunctionalization

```
(def eval (env term cont)
  (match term
    ([String x]
      (let val (env x))
      (continue1 cont val))
    ({Abs x body}
      (let val1 {Fun body env x})
      (continue1 cont val1))
    ({App fn arg} (eval env fn {App1 arg cont env}))))

(def apply (fn1 v cont1)
  (match fn1
    ({Fun body env x}
      (let var1 (extend env x v))
      (eval var1 body cont1))))

(def continue1 (fn2 var3)
  (match fn2
    ({App2 cont var2} (apply var2 var3 cont))
    ({App1 arg cont env} (eval env arg {App2 cont var3}))
    ({Halt } var3)))
```

Finishing Touches

- Inline let-bindings generated by transformation
Only if the variable is used exactly once
- Future work: sugar single branch matches as let bindings

The Result

```
(def eval (env term cont)
  (match term
    ([String x] (continue1 cont (env x)))
    ({Abs x body} (continue1 cont {Fun body env x}))
    ({App fn arg} (eval env fn {App1 arg cont env}))))

(def apply (fn1 v cont1)
  (match fn1 ({Fun body env x} (eval (extend env x v) body cont1))))

(def continue1 (fn2 var3)
  (match fn2
    ({App2 cont var2} (apply var2 var3 cont))
    ({App1 arg cont env} (eval env arg {App2 cont var3}))
    ({Halt } var3)))
```

Demo: Actually Using the Tool

call-by-value normalization by evaluation

Case studies

Language	Interpreter style	Lang. Features	Result
call-by-value λ -calculus	denotational	.	CEK machine
	denotational	integers with add	CEK with add
	denotational, recursion via environment	integers, recursive let-bindings	similar to Reynold's first-order interpreter
	denotational with conts.	shift and reset	two layers of conts.
	denotational, monadic	exceptions with handlers	explicit stack unwinding
	denotational, CPS		pointer to exception handler
	normalization by evaluation	.	strong CEK machine
call-by-name λ -calculus	big-step	.	Krivine machine
call-by-need λ -calculus	big-step (state passing)	memoization	lazy Krivine machine
simple imperative	big-step (state passing)	conditionals, while, assignment	.
micro-Prolog	CPS	backtracking, cut operator	logic engine

Conclusion

- Algorithm
 - ▶ Fully automatic transformation
 - ▶ Works with interpreters expressed in a higher-order language
 - ▶ Allows for fine-grained control over the resulting machine
- Implementation
 - ▶ Interpreters embedded in *Racket* source files
 - ▶ Users influence transformation using annotations
 - ▶ Tested on a selection of interpreters
- Current work: *Coq* formalization
 - ▶ First version: encodings of natural semantics
 - ▶ Define transformation on a deeply embedded language
 - ▶ Leverage *MetaCoq* library to transform to and from shallow embeddings

- Reynolds: *Definitional Interpreters for Higher-Order Programming Languages*
- Ager, Biernacki, Danvy and Midtgaard: *A functional correspondence between evaluators and abstract machines*
- Van Horn and Might: *Abstracting abstract machines*